
1345 – Arquitectura de Computadores

Relatório do 2º Teste formativo

   Ano lectivo 1999/2000

Resolução da questão 1 

O mapa de memória pedido é o seguinte:

[image: image4.wmf]Endereço do 

fim da página

Endereço do início 

da página

FFFFH

Página 7 

8K

E000h

DFFFH

Página 6 

8K

C000H

BFFFH

Página 5 

8K

A000H

9FFFH

Página 4 

8K

8000H

7FFFH

Página 3 

8K

6000H

5FFFH

Página 2 

8K

4000H

3FFFH

Página 1 

8K

2000H

1FFFH

Página 0 

8K

0000H

 

A partir  do mapa de memória anteriormente apresentado, se conclui que a selecção das páginas de memória de 8K é feita a custa dos 3 bits mais significativos do espaço de endereçamento. A determinação da página de 8K a seleccionar é feita de acordo com a tabela de verdade a seguir apresentada.

Outra forma de chegar a mesma conclusão, consiste em lembrar-nos de que no espaço de endereçamento de 64K somente há espaço para 8 página de memória de 8K (Como é obvio isto implica que nenhuma das paginas de 8K tem o seu espaço de endereçamento parcial/totalmente  sobreposto ao de outra(s) página(s) de 8K) .

O circuito da figura a seguir apresentada implementa a função requerida:

Foi assumida a existência dos sinais WR e RD (lógica positiva) derivados dos sinais 

 do CPU. O sinal de 

 funciona como um trinco que inibe o circuito de descodificação das páginas de memória de 8K quando o CPU pretende  aceder a um qualquer porto de entrada/saída.

Resolução da questão 2

Para a implementação da página de memória de 8Kx8 a partir dos módulos RAM  2114A, seguiremos o seguinte raciocínio: Para implementar um endereçamento de 8K são necessários 13 bits (A12-A0). Para endereçar as 2114A, 1Kx4 são necessários 10 bits (A0-A9). De aqui facilmente se conclui que os 10 bit LSB do espaço de endereçamento da página de 8K irão directamente as entradas de endereçamento das RAMs, pelo que só precisamos dos 3 bit mais significativos do espaço de endereçamento de 8K para seleccionar os diversos pares de 2114A necessários a implementação  do módulo. São precisos 8 pares de 2114A para implementar uma página de 8Kx8. As próximas figuras tornarão o raciocínio mais explícito. A entrada que faz a desinibição do circuito responsável pela descodificação dos [image: image8.png]C5_MODT

módulos de 1K na página de 8K (CS_PG#) está ligada a 1 e somente 1 das saídas do circuito de descodificação das páginas de 8K (CS_PG0...CS_PG7). Isto garante que somente o circuito que faz o mapeamento dos módulos de 1K nas páginas de 8K, pertencente a página de 8K que se encontra seleccionada, se encontra activo.

CIRCUITO RESPONSAVEL PELO MAPEAMENTO DOS MÓDULOS DE  1K (2114A) NA PÁGINA DE 8K

DIAGRAMA DE BLOCOS SIMPLIFICADO DE UMA PÁGINA DE MEMÓRIA DE 8Kx8 IMPLEMENTADA A CUSTA DE MÓDULOS DE  1Kx4 (2114A) 

Resolução da questão 3

A próxima figura apresenta  o circuito de endereçamento dos 4 portos pelo método da selecção linear.

Foi considerado a existência dos sinais de I/OR E I/OW (lógica positiva) e que as entradas das portas são igualmente activas a 1. O sinal de I/OR se encontra activo sempre que o CPU pretende fazer uma leitura sobre um porto de entrada. O sinal de I/OW pelo seu lado se encontra activo sempre que o CPU pretende fazer uma escrita sobre um porto de saída.

Resolução da questão 4

Uma solução que satisfaz os requisitos estipulados é dada pela seguinte tabela de verdade:

O circuito de descodificação, encontra-se na seguinte figura:

No que se refere aos programas pedidos, a seguir se apresentam as diversas listagens dos programas.

Resolução da questão 5

; **********************************************************

; PROGRAMA:MAIOR

; FUNÇÃO:GUARDAR NO ENDEREÇO 0060H O MAIOR DOS CONTEÚDOS DAS

; Memórias DE ENDEREÇOS ENTRE 0040H E 005FH

; CONTEUDOS RESTRINGIDOS ENTRE 0-127

.EQU     TABLE,     H'0040      ;INICIO DA TABELA DE DADOS

.EQU     RANGE,     H'01F       ;NO DE ELEMENTOS

.EQU     RES,       H'0060      ;ENDEREÇO DESTINO

.ORG  H'0000                    ;INICIO DO PROGRAMA

MAIOR:  LXI     H,TABLE         ;HL PASSA APONTAR O INICIO DA

                                ;TABELA        

        MVI     A,0             ;INICIA A COM O MENOR VALOR

                                ;POSSIVEL

        MVI     C,RANGE         ;INICIALIZA CONTADOR

LOOP:   CMP     M

        JNC     L1              ;ACC >= M[HL] ?

        MOV     A,M             ;NAO -> ACTUALIZA ACC

L1:     INX     H               ;SIM -> PASSA PARA A PROXIMA

                                ;POSICAO

        DCR     C               ;DECREMENTA CONTADOR

        JNZ     LOOP            :FIM DE LOOP?

        STA     RES             ;AFECTA RES COM O MAIOR VALOR 

                                ;ENCONTRADO

        HLT

.END

Resolução da questão 6
; **********************************************************

; PROGRAMA:ALTERE

; FUNÇÃO:ALTERA PARA 00 OS CONTEÚDOS NEGATIVOS DETECTADOS NAS 

; MEMORIAS DE ENDEREÇOS ENTRE 0030H E 0040H

; 

.EQU     TABLE,     H'0030      ;INICIO DA TABELA DE DADOS

.EQU     RANGE,     H'010       ;NO DE ELEMENTOS

.ORG  H'0000                    ;INICIO DO PROGRAMA

ALTERE: LXI     H,TABLE         ;HL APONTA PARA O INICIO DA

                                ;TABELA        

        MVI     C,RANGE         ;INICIALIZA CONTADOR

LOOP:   MOV     A,M             ;ACC <- M[HL]

        ORA     A

        JP      L1              ;ACC >= 0?

        XRA     A               ;NAO -> ACC <- 0

        MOV     M,A             ;ACTUALIZA M[HL]

L1:     INX     H               ;SIM -> INCREMENTA O APONTADOR 

;PARA A PROX.

                                ;POSICAO

        DCR     C               ;DECREMENTA CONTADOR

        JNZ     LOOP            ;FIM DE LOOP?

        HLT

.END

Resolução da questão 7

; **********************************************************

; PROGRAMA:DESLOCAR

; FUNÇÃO:DESLOCAR UM LED ACESSO NUMA FILA DE 8 LEDS, FUNÇÃO DE

; 2 INTERRUPTORES LIGADOS A UM PORTO DE SAIDA. O SENTIDO DO DES-

; LOCAMENTO DEPENDE DO INTERRUPTOR I1 E A CONTINUAÇÃO OU PARAGEM

; DEPENDE DO INTERRUPTOR I2 

; ASSUME-SE QUE O INTERRUPTOR I1 ESTA LIGADO AO BIT A0 DA PORTA

; E O INTERRUPTOR I2 AO BIT A7 DA MESMA PORTA.

; CONSIDERA-SE AINDA QUE SE NO BIT A0 O VALOR FOR 0 ENTÃO O DESLO-

; CAMENTO É PARA DIREITA SE FOR 1 PARA ESQUERDA

; CONSIDERA-SE AINDA QUE SE BIT A7 DA PORTA ESTIVER O VALOR 1 QUE O

; DESLOCAMENTO ESTA AUTORIZADO, CASO CONTRARIO QUE ESTA INIBIDO.

; PARA EFEITOS DA FILA DE LEDS, CONSIDERA-SE AINDA QUE O LED SITUADO

; NA EXTREMIDADE MAIS A ESQUERDA ESTA LIGADO AO BIT A7 DA PORTA DE

; SAIDA ENQUANTO O QUE ESTA SITUADO MAIS A DIREITA  ESTA LIGADO AO

; BIT A0

; ASSUME-SE AINDA QUE O REFRESCAMENTO É FEITO SENSIVELMENTE DE SEG.

; EM SEGUNDO, QUE EXISTE UMA ROTINA DE DELAY APROPRIADA, QUE A PORTA

; DE ENTRADA SE ENCONTRA NO ENDEREÇO 10H E A PORTA DE SAÍDA NO ENDERE-; ÇO 11H.

.EQU     STACK,      H'2000      ;INICIO  DA PILHA

.EQU     PORT_A,     H'10        ;PORTA DE ENTRADA

.EQU     PORT_B,     H'11        :PORTA DE SAIDA

.EQU     LED_INICIO, H'01        ;CONFIGURAÇÃO INICIAL DOS LEDS

.ORG  H'0000                     ;INICIO DO PROGRAMA

INICIO:         LXI     SP,STACK        ;INICIALIZA O SP

                MVI     B,LED_INICIO    ;INICIALIZAÇÃO COM A

                                        ;CONFIGURACAO

                                        ;DE INICIO

FOREVER:        CALL    AUTORIZAR       ;ROTINA QUE VERIFICA SE NO

 





     ;PORTO

                                        ;DE ENTRADA O INTERRUPTOR I2

                                        ;ESTA DESINIBIDO

                CALL    DESLOCAR        ;ROTINA QUE FAZ O DESLOCAMENTO

                CALL    DELAY           ;ROTINA DE DELAY FORNECIDA

                JMP     FOREVER

AUTORIZAR:      IN      PORT_A  ;DETERMINA SE I2 ESTA DESINIBIDO

                RAL

                RC

                JMP     AUTORIZAR ;FAZ POLLING ENQUANTO NAO ESTIVER

                                  ;I2 DESINIBIDO               

DESLOCAR:       RAR               ;DESLOCAR O LED DE ACORDO COM I1

                RAR

                JC      ESQUERDA

                MOV     A,B     ;DESLOCAMENTO PRA DIREITA

                RRC             ;(A7)<-(A0);CY<-(A0);(A[n])<- A[n+1])

                OUT     PORT_B

                MOV     B,A

                RET

ESQUERDA:       MOV     A,B     ;DESLOCAMENTO PRA ESQUERDA

                RLC             ;(A0)<-(A7);CY<-(A7);A[n+1])<-(A[n])

                OUT     PORT_B

                MOV     B,A

                RET

.END

Resolução da questão 8
Considerando  que o display é activo a 0, e tendo em conta os dados apresentados na seguinte tabela, podemos determinar o valor a ser escrito no display de 7 segmentos.

;***************************************************************

; PROGRAMA:CONTAGEM

; FUNÇÃO:COMANDAR POR RELOGIO EXTERNO, ATRAVES DA ENTRADA 

; RST6.5 O RITMO A QUE É INCREMENTADO UM DIGITO QUE É

; VISUALIZADO NUM DISPLAY 7 SEGMENTOS (0->1->2->..9->0->..) 

; TENDO EM CONTA QUE A ENTRADA RST 6.5 É ACTIVADO POR NÍVEL, 

; TER-SE-A QUE SIMULAR POR SOFTWARE Á ACÇÃO ACTIVADO POR FLANCO

.EQU     STACK,                 H'2000      ;INICIO  DA PILHA

.EQU     PORT_B,                H'11        :PORTA DE SAIDA

.EQU     DISPLAY_INICIO,        H'00        ;CONFIGURACAO

 







 ;INICIAL DOS LEDS

.ORG  H'0000                    ;INICIO DO PROGRAMA

INICIO:         LXI     SP,STACK ;INICIALIZAÇÃO DO SP

                MVI     A, H'0D  ;ACC <- 00001101

                SIM              ;DESMASCAR O RST6.5

                MVI     B,DISPLAY_INICIO  ;INICIALIZA O CONTADOR 

                                          ;DE DIGITOS

LOOP1:          CALL    CONVERSAO_DYS     ;ACTUALIZA DISPLAY DE 

                                          ;7 SEGMENTOS COM O

                                          ;VALOR DO DIGITO

                                          ;ARMAZENADO NO REG B

                

                EI

                HLT                       ;AGUARDA ACÇÃO DE

                                          ;INTERRUPT

INR     B

                MOV     A,B

                CPI     H'0A              ;CHEGAMOS A 10?

                JM      LOOP1             ;NAO -> CONTINUA LOOP

                MVI     B,DISPLAY_INICIO  ;SIM ->FAZ

                                          ;INICIALIZAÇÃO AO

                                          ;CONTADOR

                JMP     LOOP1             ;REPETE O CICLO SEM

                                          ;FIM

;***************************************************************

;A ROTINA CONVERSAO_DYS EXECUTA UMA TABLE LOOKUP, UTILIZANDO 

;COMO INDICE PARA A TABELA O ENDEREÇO CONSTITUIDO PELA BASE (O 

;APONTADOR [HL] QUE CONTEM O INICIO DA TABELA) E UM OFFSET 

; CONSTITUIDO POR UM APONTADOR DE 2 BYTES (PAR DE), NO QUAL O ;REGISTO [E] E AFECTADO COM O VALOR DO DIGITO A SER VISUALIZADO

CONVERSAO_DYS:  LXI     H,TABLE_SEG

                LXI     D,H'0

                MOV     A,B

                MOV     E,A

                DAD     D

                MOV     A,M

                OUT     PORT_B

                RET

.ORG     H'34

; ROTINA DE ATENDIMENTO DO RST 6.5

RST65:          PUSH    PSW     ;SALVAR TODOS OS REGISTO DO CPU

                                ;NO STACK

                PUSH    B

                PUSH    D

                PUSH    H

INNER:          RIM             ;RECOLHER NO ACC A IMAGEM DO RST 

                                ;QUE REQUEREM ATENDIMENTO

                ANI     H'20    ;EXTRAIR EXCLUSIVAMENTE

 





;INFORMAÇÃO

                                ;SOBRE A ENTRADA RST 6.5

                JNZ     INNER   ;ENQUANTO ESTIVER ACTIVA

 




      ;AGUARDAR. ESTA SENDO SIMULADA







;POR SOFTWARE A ACÇÃO ACTIVADO 







;POR FLANCO

                POP     H       ;RESTAURAR TODOS OS REGISTO DO

; CPU A 

                POP     D       ;PARTIR DO STACK

                POP     B

                POP     PSW     

                RET

TABLE_SEG:      .DB      H'01   ; DIGITO 0

                .DB      H'4F   ; DIGITO 1

                .DB      H'12   ; DIGITO 2

                .DB      H'06   ; DIGITO 3

                .DB      H'4C   ; DIGITO 4

                .DB      H'24   ; DIGITO 5

                .DB      H'60   ; DIGITO 6

                .DB      H'0F   ; DIGITO 7

                .DB      H'00   ; DIGITO 8

                .DB      H'0C   ; DIGITO 9

.END

